

## A Note on a Reformulation of the KHB Method

Richard Breen, Department of Sociology and Nuffield College, University of Oxford

Kristian Karlson, Department of Sociology, Copenhagen University

Anders Holm, Department of Sociology, University of Western Ontario

**Abstract:** The KHB method has rapidly become popular as a way of separating the impact of confounding from rescaling when comparing conditional and unconditional parameter estimates in non-linear probability models like the logit and probit. In this note we show that the same estimates can be obtained in a somewhat different way to that advanced by Karlson, Holm, and Breen (2012) in their original article and implemented in the user-written *Stata* command *khb*. While the KHB method and this revised KHB method both work by holding constant the residual variance of the model, the revised method makes comparisons across multiple nested models easier than the original method.

### **Introduction**

Sociologists and other social scientists routinely compare coefficients across same-sample nested models. For instance, when sociologists want to investigate how much of the association between a variable of interest,  $X$ , and an outcome,  $Y$ , is mediated through a third variable,  $Z$ , they often first fit a model with  $Y$  as the dependent variable and  $X$  as the predictor. Then they fit another model that adds  $Z$  as a control variable. The coefficient for  $X$  from this model is compared with the same coefficient from the first model. The difference between the two is then taken as a measure of how far the relationship between  $X$  and  $Y$  is mediated by  $Z$ .

If the models used are linear (estimated by, for example, ordinary least squares) this procedure is unproblematic, but this is not so for non-linear models like the logit, probit, multinomial logit or ordered logit, which would often be used if the outcome,  $Y$ , were categorical or ordinal. Because non-linear models are non-

collapsible over confounders (or mediators), the change in the coefficient of the independent variable of interest is not solely driven by mediating or confounding by the added variables; part of the change is due to change in the residual variance of the models. The more variables that are included in a non-linear probability model, the smaller the residual variance and, because the variance and mean are not separately identified in these models, change in the residual variance also affects the coefficient estimates. One consequence of this is that, even if a variable unrelated to  $X$  is added to the model, the coefficient of  $X$  will change.<sup>1</sup>

Several methods have been suggested to separate this so-called “re-scaling effect” from the true degree of confounding or mediation. They include Y-standardisation, average marginal effects, and, more recently, the KHB method. The KHB method compares the regression coefficients for the variable of interest  $X$  from two models. One includes  $X$ , and the mediators/confounders,  $Z$  (which may be a single variable or a vector) and one includes  $X$  and a residualized version of  $Z$ . The  $Z$  variables are residualized with respect to  $X$ . Both models have the same predictive power and hence the same residual variance, but the residualized  $Z$  are uncorrelated with  $X$  and so a comparison of coefficients of the independent variable of interest across these models reveals the true impact of mediation/confounding on the coefficient for  $X$ .

In this paper, we present a simple method to obtain exactly the same measure of confounding/mediation, using the latent index (that is, the estimated linear predictor) from a non-linear probability model. We show that this approach allows the researcher to infer the amount of mediation/confounding through simple linear regression analyses. We illustrate it with an example using the ordered logit model.

## **The method**

---

<sup>1</sup> Further explanation and proofs of this can be found in Karlson, Holm and Breen (2012) and Breen, Holm and Karlson (2013). The literature dealing with this issue is reviewed in Breen, Karlson and Holm (forthcoming).

We begin by defining the following latent linear regression models

$$Y^* = \beta_0 + \beta_1 X + \beta_2 Z + \varepsilon \quad (1a) \text{ this is the full model}$$

$$Y^* = \gamma_0 + \gamma_1 X + \nu \quad (1b) \text{ this is the reduced model}$$

$Y^*$  is a continuous, but unobserved, latent dependent variable,  $X$  is the independent variable of interest,  $Z$  is a mediator or confounder,  $\varepsilon$  is an error term assumed to be independent of  $X$  and  $Z$  and  $\nu$  is an error term assumed independent of  $X$ .  $Z$  may be a single variable or a vector of variables.

We observe the binary indicator,  $Y$ , defined in terms of the unobserved  $Y^*$  by

$$Y = 1 \text{ if } Y^* > \tau$$

$$Y = 0 \text{ if } Y^* \leq \tau$$

with  $\tau$  being an unknown threshold.

The non-linear regressions corresponding to the latent linear models are

$$h(\text{pr}(Y = 1)) = b_0 + b_1 X + b_2 Z \quad (2a)$$

$$h(\text{pr}(Y = 1)) = c_0 + c_1 X \quad (2b)$$

where  $h(\dots)$  is a non-linear transformation such as the logit or probit. We know that the relationship between the coefficients of the non-linear regression, (2a) and the latent variable model, (1a) is  $b_j = \frac{\beta_j}{s}, j = 1, 2$ , where  $s$  is a scale factor that allows the variance of the error term in (1a) to differ from that of an assumed standard distribution such as the standard logistic or standard normal. The same holds for the relationship between the coefficients in equations (2b) and (1b), albeit with a different scale factor (since this reflects the residual or unexplained variance of the model).

The KHB method works by residualizing the mediator/confounder,  $Z$ , by regressing it on  $X$  using ordinary least squares, and then another non-linear model is estimated:

$$h(\text{pr}(Y = 1)) = a_0 + a_1X + a_2\tilde{Z} \quad (3)$$

where  $\tilde{Z}$  is the residualized  $Z$ . This corresponds to an underlying latent model:

$$Y^* = \alpha_0 + \alpha_1X + \alpha_2\tilde{Z} + u \quad (4)$$

where  $u$  is an error term assumed to be independent of  $X$  and  $Z$ . Karlson, Holm, and Breen (2012) show that  $\varepsilon = u$  (and thus their variances and the implied scale factors are the same) and that  $\alpha_1 = \gamma_1$ . It then follows that the KHB estimator,  $\frac{b_1}{a_1}$  is equal to  $\frac{\beta_1}{\gamma_1}$ : that is, the ratio of the conditional (partial) to the unconditional (gross) coefficient of  $X$  in the underlying latent model.

The revised method presented here recovers the same ratio but in an easier way. Define  $V$  as the linear predictor (latent index) from model (2a); that is

$$V = h^{-1}(\text{pr}(Y = 1)) = b_0 + b_1X + b_2Z \quad (5)$$

Next, we fit two OLS models, using  $V$  as the dependent variable:

$$V = \theta_0 + \theta_1X + \theta_2Z \quad (6a)$$

$$V = \varphi_0 + \varphi_1X + \omega \quad (6b)$$

Model (6a) has no error term because it is saturated: it fits the data perfectly.

Karlson, Holm and Breen (2012) proved that the KHB method recovers the ratio

$\frac{\beta_1}{\gamma_1}$ . From equations (1a) and (1b) this ratio is equal to:

$$\frac{\beta_1}{\gamma_1} = \frac{\beta_1}{\beta_1 + \beta_2 \text{cov}(Z, X) / \text{var}(X)}$$

Thus we need to show that revised method, using (6a) and (6b), recovers the same ratio: that is, we need to show that  $\frac{\theta_1}{\varphi_1} = \frac{\beta_1}{\gamma_1}$ .

For all coefficients, we know that  $\theta_j = b_j$  holds; that is, the coefficients from the linear model fitted to the latent index are the same as the coefficients that generated the latent index. Then

$$\varphi_1 = \theta_1 + \theta_2 \frac{\text{cov}(Z, X)}{\text{var}(X)} = b_1 + b_2 \frac{\text{cov}(Z, X)}{\text{var}(X)}$$

and this yields:

$$\frac{\theta_1}{\varphi_1} = \frac{b_1}{b_1 + b_2 \frac{\text{cov}(Z, X)}{\text{var}(X)}} = \frac{\beta_1 / s}{\frac{1}{s} \left[ \beta_1 + \beta_2 \frac{\text{cov}(Z, X)}{\text{var}(X)} \right]} = \frac{\beta_1}{\gamma_1}$$

This method returns exactly the same ratio estimates of confounding or mediation as the KHB method and these are equal to the true ratio in the underlying latent variable model in equations (1a) and (1b).

### **Comparing the revised method with the original KHB**

The method proposed here requires the following steps:

- 1 fit the full model, including  $X$  and all other predictors of  $Y$ , as a logit or other non-linear probability model;
- 2 save the latent index from this model;
- 3 taking the latent index as the dependent variable, use OLS models (with and without the mediators/confounders), to estimate the extent of mediation or confounding.

This steps for the original KHB method are:

- 1 regress the confounder on the independent variable of interest;
- 2 obtain the predicted values from this regression;
- 3 fit the full model;
- 4 fit the reduced model, controlling for the predicted values..

At first sight it may seem that both methods are equally convenient, but suppose we wanted to introduce mediating or confounding variables sequentially (that is, through successive models, starting with a model including  $Z_1$ , followed by a model including  $Z_1$  and  $Z_2$ , and so on) to see where the coefficient of  $X$  changed most. In this case the method presented here would be more convenient because it would not require any extra steps other than an additional OLS regression including both  $Z_1$  and  $Z_2$ . Using the KHB method, however, we would need to residualize both of these, using separate regressions. As the number of mediators or confounders whose impact we want to assess increases, so does the computational advantage of the new method.

### Standard errors

Karlson, Holm and Breen (2012: 295-6) developed a Z-test for whether the conditional and unconditional effects of  $X$  are equal (this is labelled  $Z_c$ ). It is a test of whether the coefficient of  $X$  is changed by true confounding or mediation, net of the effect of rescaling. In our notation this test would be written:

$$Z_c = \frac{a_1 - b_1}{s.e.(a_1 - b_1)} \quad (7)$$

This is equivalent to a test of the significance of the product  $b_2\psi_1$  where  $\psi_1$  is the estimated coefficient from the equation used to residualize  $Z$ :

$$Z = \psi_0 + \psi_1 X + \omega \quad (8)$$

Karlson et al (2012) use the delta method to derive the standard error of  $b_2\psi_1$ , and thus the Z-test of the significance of the difference between the conditional and unconditional coefficients of  $X$  measured on the scale of the full model.

As an alternative, the bootstrap is a fast and convenient way of obtaining standard errors (the analytical standard errors based on the delta method can be cumbersome to calculate without specialized software).<sup>2</sup> Another advantage of using the bootstrap for estimating standard errors is that one can easily compute standard errors of differences of coefficients, i.e., of indirect effects, or of ratios of coefficients such as the percentage mediated. In this case, we would often care about our estimate of the ratio  $\frac{\theta_1}{\varphi_1}$ . To test whether it differed significantly from

one we would need to calculate its standard error, and this may be awkward, not only because it is a ratio of two estimated parameters, but also because the method presented here uses another estimated quantity – namely  $V$  – in its derivation. Because  $V$  is estimated from the data, the uncertainty in that estimate must also be taken into account in estimating the standard error of  $\frac{\theta_1}{\varphi_1}$ . Both

these issues are easily dealt with using the bootstrap, but we need to bootstrap not just the OLS regressions used to estimate  $\theta_1$  and  $\varphi_1$  but also the non-linear probability model used to generate  $V$ . So the bootstrap steps are: estimate model (2a), use this to calculate  $V$  (equation 5), fit models (6a) and (6b), and save the estimated ratio  $\frac{\theta_1}{\varphi_1}$ . The same sequence applies to get the bootstrap standard errors for any other quantities of interest.<sup>3</sup>

### **Example: Parental Income and Support for Redistribution**

To illustrate how our method works, we examine potential mediators of the association between parental income and individuals' stated support for

---

<sup>2</sup> The bootstrap is fast and convenient from the researcher's perspective because of its flexibility. In terms of computational time, however, it is not faster than analytically based estimators of the standard error. We thank an anonymous reviewer for pointing out this qualification.

<sup>3</sup> We provide an example of such a bootstrap loop in *Stata* in the appendix.

redistribution in the US using the 1987, 1993, 1994, 2008, and 2010 waves in the General Social Surveys (GSS). The final sample after listwise deletion is 3,603.

The outcome variable is the respondents' response to a question about whether the government should reduce income differentials, and has five ordered response categories: strongly disagree, disagree, neither, agree, strongly agree. The main predictor variable is the respondent's parents' income when the respondent was 16 years old. We recode this variable into three levels: below average, at average, above average. As mediators we include socioeconomic and sociodemographic characteristics of the respondent: the respondent's income grouped into 12 income bins (which we use as a continuous covariate), the respondent's years of schooling, the respondent's occupational prestige score (2010 scoring on a 100-point scale), the marital status of the respondent (married, widowed, divorced, separated, never married), and the respondent's region of residence (New England, Middle Atlantic, East North Central, West North Central, South Atlantic, East South Central, West South Central, Mountain, Pacific). In addition to these variables, we control all models for survey year fixed effects and the respondent's age to balance the sample on these characteristics.

Table 1 shows the estimated association, using ordered logit models, between parental income and a respondent's support for government action to reduce income differentials. The gross or unmediated association in Model M1 suggests that respondents born to affluent parents less often support redistribution (as defined here). Introducing respondent characteristics as mediators leads to a decline in the effect of parental income across the models, particularly across the first three. For example, the coefficient for above-average parental income declines by 24.7% from Model M1 (-0.673) to Model M3 (-0.507).

[Table 1 here]

Rescaling may influence comparisons of the effect of parental income across these same-sample nested models. Therefore, in Table 2, we report the

corresponding estimates using the revised KHB approach detailed in this paper. It appears that, in fact, changes in the coefficients are not substantively driven by rescaling: the rescaling effect is negligible, as can be seen by comparing the gross association in the first columns in Tables 1 and 2. For example, the coefficient for above-average parental income declines by 26.2 percent from Model M1 (-0.693) to Model M3 (-0.511). This relative decline is slightly larger than the one reported in Table 1, but would not lead to meaningfully different conclusions about the degree to which respondent attainment explains or mediates the association between parental income and support for redistribution.

[Table 2 here]

The standard errors for the coefficients for parental income reported in Table 2 are estimated using the bootstrap. In Table 3 we give an example and we report our *Stata* code for the bootstrap loop in the appendix. We report estimates of the indirect effects and the corresponding percentage mediated involving Models M1, M2, and M3 in Table 2. We find that all indirect effects are statistically significant at conventional levels, suggesting that the respondent's socioeconomic attainment is a significant mediator of the association between parental income and support for redistribution. Adding the respondent's income to the model (M2) reduces the effects by about 8% and 10% for respondents whose parents were at the average income and respondents whose parents exceeded the average income, respectively. Further adding respondent's years of schooling results in about 17% and 26% mediated, respectively. Also, the incremental mediation by respondent's years of schooling over and above that by the portion mediated by respondent's income is about 9% and 19%, respectively. All of these mediation percentages are statistically significant at conventional significance levels, again supporting the conclusion that a respondent's socioeconomic characteristics explain some, but not all, of the association.

[Table 3 here]

## Conclusion

The KHB method has rapidly become popular. In this note we have provided, and illustrated, a reformulation of the method, which, we believe, is simpler to implement than that used in the original paper and implemented in the *Stata* program, *khb*<sup>4</sup>, Kohler, Karlson, and Holm (2011). It has particular advantages when we want to consider how the association between a categorical or ordered outcome and a predictor variable of interest changes when we add several mediators or confounders sequentially.

---

<sup>4</sup> For a similar package written for R, see <https://www.rdocumentation.org/packages/matchingMarkets/versions/0.3-2/topics/khb>

## References

Breen, Richard, Kristian Bernt Karlson, and Anders Holm. 2013. "Total, Direct, and Indirect Effects in Logit and Probit Models." *Sociological Methods and Research* 42: 164-91.

Breen, Richard, Kristian Bernt Karlson, and Anders Holm. 2018. "Interpreting and Understanding Logits, Probits, and other Non-Linear Probability Models." *Annual Review of Sociology* 44.

Karlson, Kristian Bernt, Anders Holm, and Richard Breen. 2012. "Comparing Regression Coefficients between Same-Sample Nested Models using Logit and Probit: A New Method." *Sociological Methodology* 42: 274-301.

Kohler, Ulrich, Karlson, Kristian Bernt, Anders Holm. 2011. "Comparing coefficients of nested nonlinear probability models." *The Stata Journal* 11: 420-438.

## TABLES

**Table 1 Ordered Logit Models Regressing Support for Redistribution on Parental and Respondent Characteristics. Logit Coefficients and Selected Standard Errors in Parentheses.**

	M1	M2	M3	M4	M5
<u>Parental Income</u>					
Below avg. (ref.)					
At avg.	-0.378 (0.069)	-0.353 (0.070)	-0.322 (0.070)	-0.320 (0.070)	-0.321 (0.070)
Above avg.	-0.673 (0.088)	-0.618 (0.088)	-0.507 (0.090)	-0.506 (0.090)	-0.515 (0.090)
<u>Respondent Characteristics</u>					
Income		-0.0784	-0.0642	-0.0601	-0.0580
Years of Schooling			-0.0661	-0.0559	-0.0575
Occupational Prestige				-0.00503	-0.00385
Marital Status Controlled	No	No	No	No	Yes
Region Controlled	No	No	No	No	Yes
<u>Thresholds</u>					
Threshold 1	-2.985	-3.532	-4.305	-4.334	-4.079
Threshold 2	-1.320	-1.848	-2.607	-2.636	-2.369
Threshold 3	-0.414	-0.929	-1.684	-1.712	-1.439
Threshold 4	1.457	0.960	0.208	0.181	0.464

Note: All models controlled for survey year fixed effects and respondent's age using a linear age specification.

**Table 2 Rescaled Logit Models using Linear Predictor Method, Regressing Support for Redistribution on Parental and Respondent Characteristics. Logit Coefficients Measured on the Scale of the Full Model and Selected Bootstrapped Standard Errors in Parentheses (500 replications)**

	M1	M2	M3	M4	M5
<u>Parental Income</u>					
Below avg. (ref.)					
At avg.	-0.390 (0.071)	-0.359 (0.071)	-0.326 (0.071)	-0.324 (0.071)	-0.321 (0.071)
Above avg.	-0.693 (0.094)	-0.627 (0.093)	-0.511 (0.095)	-0.510 (0.095)	-0.515 (0.094)
<u>Respondent Characteristics</u>					
Income		-0.081	-0.066	-0.062	-0.058
Years of Schooling			-0.067	-0.057	-0.058
Occupational Prestige				-0.005	-0.004
Marital Status Controlled	No	No	No	No	Yes
Region Controlled	No	No	No	No	Yes

Note: All models controlled for survey year fixed effects and respondent's age using a linear age specification.

**Table 3 Selected Indirect Effects and Percent Mediated of Parental Income Effect on Support for Redistribution. Logit Coefficients, Percent Mediated, and Bootstrapped Standard Errors in Parentheses (500 replications)**

	Parental Income	
	At avg.	Above avg.
Indirect Effects		
M1 → M2	-0.031 (0.011)	-0.066 (0.015)
M1 → M3	-0.064 (0.015)	-0.182 (0.028)
M2 → M3	-0.033 (0.010)	-0.116 (0.023)
Percent Mediated		
M1 → M2	8.1% (3.3)	9.5% (2.4)
M1 → M3	16.5% (5.2)	26.2% (5.3)
M2 → M3	9.2% (3.6)	18.5% (4.7)

Note: Based on coefficients reported in Table 2. Reference category for Parental Income is Below Average.

## Appendix

*Stata dofile replicating results reported in empirical example*

```
use gss_redist.dta, clear // GSS data (selected variables)
```

\* Listwise deletion

```
egen miss = rmiss(redincdiff pinc year age rincome educ prestg10 marital
region)
keep if miss == 0
```

\* Descriptives

```
d goveqinc redincdiff pinc year age rincome educ prestg10 marital region
tab1 goveqinc redincdiff pinc year age rincome educ prestg10 marital region
```

\* Naive Approach

```
ologit redincdiff b1.pinc i.year c.age
est store n1
ologit redincdiff b1.pinc i.year c.age c.rincome
est store n2
ologit redincdiff b1.pinc i.year c.age c.rincome c.educ
est store n3
ologit redincdiff b1.pinc i.year c.age c.rincome c.educ c.prestg10
est store n4
ologit redincdiff b1.pinc i.year c.age c.rincome c.educ c.prestg10 i.marital
i.region
est store n5
est tab n1 n2 n3 n4 n5 , se // Table of naive models
```

\* Linear Predictor Approach

```
// Step 1: Run "Full" model (including all covariates) and save linear predictor
ologit redincdiff b1.pinc i.year c.age c.rincome c.educ c.prestg10 i.marital
i.region
predict xb, xb
// Step 2: Run reduced models on linear predictor to construct table of models
reg xb b1.pinc i.year c.age
est store m1
reg xb b1.pinc i.year c.age c.rincome
est store m2
reg xb b1.pinc i.year c.age c.rincome c.educ
est store m3
reg xb b1.pinc i.year c.age c.rincome c.educ c.prestg10
est store m4
reg xb b1.pinc i.year c.age c.rincome c.educ c.prestg10 i.marital i.region
est store m5
est tab m1 m2 m3 m4 m5, not nostar // Table of "correct" models
```

\* Bootstrap SE for Linear Predictor Approach

```
cap progr drop bsxb // erase any stored programs
progr define bsxb, rclass // program starts
```

```

// Step 1: Run "Full" model and save linear predictor
ologit redincdiff b1.pinc i.year c.age c.rincome c.educ c.prestg10 i.marital
i.region
cap drop xb
predict xb, xb
// Step 2: Run models and save parameters of interest relevant for inference
reg xb b1.pinc i.year c.age
return scalar m1_at = _b[2.pinc]
return scalar m1_above = _b[3.pinc]
reg xb b1.pinc i.year c.age c.rincome
return scalar m2_at = _b[2.pinc]
return scalar m2_above = _b[3.pinc]
reg xb b1.pinc i.year c.age c.rincome c.educ
return scalar m3_at = _b[2.pinc]
return scalar m3_above = _b[3.pinc]
reg xb b1.pinc i.year c.age c.rincome c.educ c.prestg10
return scalar m4_at = _b[2.pinc]
return scalar m4_above = _b[3.pinc]
reg xb b1.pinc i.year c.age c.rincome c.educ c.prestg10 i.marital i.region
return scalar m5_at = _b[2.pinc]
return scalar m5_above = _b[3.pinc]
drop xb
end // program ends
// Run bootstrap program:
bs m1_at=r(m1_at) m1_above=r(m1_above) ///
m2_at=r(m2_at) m2_above=r(m2_above) ///
m3_at=r(m3_at) m3_above=r(m3_above) ///
m4_at=r(m4_at) m4_above=r(m4_above) ///
m5_at=r(m5_at) m5_above=r(m5_above) ///
, reps(500) seed(1001): bsxb, all

```

\* Bootstrap SE for Linear Predictor Approach: Indirect Effects and Percent Mediated

```

cap progr drop bsxbdif // erase any stored programs
progr define bsxbdif, rclass // program starts
// Step 1: Run "Full" model and save linear predictor
ologit redincdiff b1.pinc i.year c.age c.rincome c.educ c.prestg10 i.marital
i.region
cap drop xb
predict xb, xb
// Step 2: Run models and save parameters of interest
reg xb b1.pinc i.year c.age
scalar m1_at = _b[2.pinc]
scalar m1_above = _b[3.pinc]
reg xb b1.pinc i.year c.age c.rincome
scalar m2_at = _b[2.pinc]
scalar m2_above = _b[3.pinc]
reg xb b1.pinc i.year c.age c.rincome c.educ

```

```

scalar m3_at = _b[2.pinc]
scalar m3_above = _b[3.pinc]
reg xb b1.pinc i.year c.age c.rincome c.educ c.prestg10
scalar m4_at = _b[2.pinc]
scalar m4_above = _b[3.pinc]
reg xb b1.pinc i.year c.age c.rincome c.educ c.prestg10 i.marital i.region
scalar m5_at = _b[2.pinc]
scalar m5_above = _b[3.pinc]
drop xb
// Step 3: Generate parameters of interest relevant for inference
return scalar m1m2_at = m1_at-m2_at
return scalar m1m2_above = m1_above-m2_above
return scalar m1m3_at = m1_at-m3_at
return scalar m1m3_above = m1_above-m3_above
return scalar m2m3_at = m2_at-m3_at
return scalar m2m3_above = m2_above-m3_above
return scalar p_m1m2_at = (m1_at-m2_at)/m1_at * 100
return scalar p_m1m2_above = (m1_above-m2_above)/m1_above * 100
return scalar p_m1m3_at = (m1_at-m3_at)/m1_at * 100
return scalar p_m1m3_above = (m1_above-m3_above) / m1_above * 100
return scalar p_m2m3_at = (m2_at-m3_at)/m2_at * 100
return scalar p_m2m3_above = (m2_above-m3_above)/m2_above * 100
end // program ends
// Run bootstrap program:
bs      m1m2_at=r(m1m2_at) ///
        m1m3_at=r(m1m3_at) ///
        m2m3_at=r(m2m3_at) ///
        p_m1m2_at=r(p_m1m2_at) ///
        p_m1m3_at=r(p_m1m3_at) ///
        p_m2m3_at=r(p_m2m3_at) ///
        m1m2_above=r(m1m2_above) ///
        m1m3_above=r(m1m3_above) ///
        m2m3_above=r(m2m3_above) ///
        p_m1m2_above=r(p_m1m2_above) ///
        p_m1m3_above=r(p_m1m3_above) ///
        p_m2m3_above=r(p_m2m3_above) ///
        , reps(500) seed(1001): bsxbdif, all

```